

Processing Kodu

Radar Projesi - Görsel Arayüz Yazılımı (Java)

```
import processing.serial.*;

Serial port;
String veri = "";
float aci = 0, mesafe = 0;
int sonNoktaZamani = 0;
ArrayList<CisimIz> izler = new ArrayList<CisimIz>();

void setup() {
    fullScreen(); // Tam ekran modu
    smooth();
    port = new Serial(this, "COM5", 9600);
    port.bufferUntil('.');
}

void draw() {
    fill(0, 15, 0, 80);
    noStroke();
    rect(0, 0, width, height);
    radarCizgileri();
    radarTaramaCizgisi();

    for (int i = izler.size()-1; i >= 0; i--) {
        CisimIz iz = izler.get(i);
        iz.goster();
        if (iz.olduMu()) izler.remove(i);
    }

    if (mesafe > 0 && mesafe < 40) {
        if (millis() - sonNoktaZamani > 1000) {
            izler.add(new CisimIz(aci, mesafe));
            sonNoktaZamani = millis();
        }
    }
    bilgiYaz();
}

void radarCizgileri() {
    stroke(0, 80, 0); strokeWeight(2); noFill();
    int merkez = width / 2;
    int taban = height - 100;
    float radarCap = height * 0.8;
    for (int r = 1; r <= 4; r++) {
        float yaricap = (radarCap / 4) * r;
```

```

        arc(merkez, taban, yaricap * 2, yaricap * 2, PI, TWO_PI);
    }
}

void radarTaramaCizgisi() {
    stroke(30, 255, 30); strokeWeight(3);
    int merkez = width / 2, taban = height - 100;
    float radarCap = height * 0.8, rad = radians(aci);
    line(merkez, taban, merkez - radarCap * cos(rad), taban - radarCap *
sin(rad));
}

void bilgiYaz() {
    fill(30, 255, 30); textSize(24);
    text("Aci: " + (int)aci, 50, 50);
    text("Mesafe: " + (int)mesafe + " cm", 50, 90);
    text("Cikis icin 'ESC'ye bas", 50, 130);
}

void serialEvent(Serial p) {
    veri = p.readStringUntil('.');
    if (veri != null) {
        veri = trim(veri);
        int virgul = veri.indexOf(',');
        if (virgul > 0) {
            aci = float(veri.substring(0, virgul));
            mesafe = float(veri.substring(virgul + 1));
        }
    }
}

class CisimIz {
    float x, y, m, parlaklik = 255;
    CisimIz(float a, float d) {
        float olcek = (height * 0.8) / 40.0;
        float rad = radians(a);
        x = (width/2) - (d * olcek) * cos(rad);
        y = (height-100) - (d * olcek) * sin(rad);
        m = d;
    }
    void goster() {
        stroke(255, 255, 0, parlaklik); strokeWeight(15);
        point(x, y);
        fill(255, 255, 255, parlaklik); textSize(18);
        text((int)m + "cm", x + 15, y - 15);
        parlaklik -= 2.1;
    }
    boolean olduMu() { return parlaklik <= 0; }
}

```